Notes on the Altos 586 Computer

Work in progress, April 2023 by Lubomir Rintel

I've recently obtained the system board from the Altos 586 computer. Unfortunately, it came without documentation, the hard drive or the controller board.

With the goal of putting the system to some use I've decided to examine the hardware and firmware, learn how the system is put together and perhaps eventually substitute the missing parts.

This document contains some notes from my investigation, provided in hope of being useful to other owners of Altos 586 systems.

1. Altos 586 vs. 586T

There seems to be some documentation available for a related system, Altos 586T. While there's no service manuals or schematics, there's a reference manual¹ that describes many aspects of the computer architecture and firmware itnerfaces.

The Altos 586 system seems to be largely similar to Altos 586T. The "T" in the latter system's designation seems to stand for "Tape", as, unlike the 586, 586T includes a tape drive.

The disk controller board used in 586T seems to be vastly different from the controller board used in the Altos 586. While the Altos 586 board is bases on around the 8089 I/O controller and only controls a hard drive, the 586T board includes a Floppy and Tape controllers and utilizes a Z80 processor to handle the I/O.

Considering the above, the 586T manual is largely relevant to the 586, except for when it comes to disk I/O, be it floppy access or hard drive access.

 $^{^{\}rm 1}$ 586T/986T System Reference Manual, P/N 690-15813-002, April 1985

2. Z80 Peripheral Controller

As on 586T, the peripheral controller is a Z80 controlling a bunch of serial interfaces, timers for baud rate generators and a RTC. It's equipped with a PIO for generating utility signal lines.

On a 586 it also has an extra duty of controlling the floppy drives, described below.

2.1. Peripherals

The peripherals are mostly the same as in the 586T machine aside from the addition of the Floppy Controller and the related circuitry.

Consult the 586T/986T System Reference Manual for description of the peripherals common to 586 and 586T.

2.2. I/O Port Map

Address	Name	Part	Description
00h			Bus Address Top Bits Latch
20h – 23h	PIT 0	Intel 8254	Programmable Interval Timer
24h – 24h	PIT 1	Intel 8254	Programmable Interval Timer
28h – 2Bh	SIO 0	Zilog Z8440APS (Z80A SIO/0)	Serial I/O Controller
2Ch – 2Fh	SIO 1	Zilog Z8440APS (Z80A SIO/0)	Serial I/O Controller
30h – 33h	SIO 2	Zilog Z8440APS (Z80A SIO/0)	Serial I/O Controller
34h – 37h	PIO	Zilog Z8420APS (Z80A PIO)	Parallel I/O Controller
38h – 3Bh	FDC	Western Digital FDC1797	Floppy Disk Controller/Formatter
3Ch	DMA	Zilog Z8410APS (Z80A DMA)	Direct Memory Access Controller
40h			DMA Carrier/Parity
80h – 9Fh	RTC	MM58167AN	Microprocessor Real Time Clock

2.3. Floppy Disk Controller/Formatter

The floppy disk controller is the FDC1797 part hooked on in a straightforward manner. The output control signals are interfaced to a standard Shuggart floppy connector via open-collector drivers, the input signals go through a line drivers with external pull up register array. The read data signal is passed through external data separator circuitry to generate clock and data from MFM encoded signal.

Some of the FDC lines involved in the Floppy Drive access are connected to the PIO. In particular, the #DDEN pin needs to be set low in order to read double-density (MFM) disks, which is almost certainly the case. The INTRQ pin is connected to PIO's PA7 input pin. No idea if it's actually used.

To actually access the floppy drive, the appropriate *Drive Select* signal needs to be pulled low. For reading double-sided floppies, the drive head/side is selected by the *Side Select* signal. The *Drive Select* and *Side Select* signals are also both controlled by the PIO.

For the details about the control lines consult the paragraph on the Parallel I/O Controller below.

2.4. Parallel I/O Controller

For the most part, the pins I've traced out are related to the floppy circuitry. Others are unknown.

#	Pin	Direction	Description
15	PA0		
14	PA1		
13	PA2		
12	PA3		
10	PA4		
9	PA5		
8	PA6		
7	PA7	Input	FDC INTRQ (Pin 39)
27	PB0	Output	"586 sync" sub-board (Pin 7)
28	PB1	Output	"586 sync" sub-board (Pin 8)
29	PB2	Input	Goes into unpopulated connector next to FDD (Pin 10)
30	PB3	Output	Floppy Drive Select 0
31	PB4	Output	Floppy Drive Select 1
32	PB5	Output	Floppy Side Select
33	PB6	Output	FDC #DDEN (Pin 37)
34	PB7	Input	PIT ? OUT2 (Pin 17)

3. Floppy Interface

3.1. Geometry and Format

The Altos 586 computer uses 51/4" diskettes in a drive with following parameters:

Data density	Double Density	
Encoding	MFM	
Speed	300 RPM	
Bit rate	250 kbps	
Tracks	80	
Heads/Sides	2	

The floppies XENIX, CP/M and the firmware use the following format (though the controller permits other formats).

Sector size	512 bytes
Sectors per track	9

Note that while this format is fairly unusual for 5¼" drives and media, it's exactly the same as IBM PC 3½" DD (720 KB) disks use. Regular 3½" drives can be used if they're adapted for the Altos 586 floppy connector pinout (see below).

The regular 1.44 MB floppies can usually be used with DD format, but the density select window needs to be covered with a piece of opaque tape.

3.2. Floppy Connector Pinout

The floppy connector is not PC compatible and therefore usual PC floppy drives can't be used without modifications. It uses straight ribbon cables, not the PC-style ones with a twist.

The pinout (and drive geometry) is fairly close to to what Commodore Amiga computers use. There seems to be a plenty of tutorials on modifying PC drives for Amiga on the Web. The same modificactions should work for the Altos 586.

Function	P	in	Direction	Function	Note
	1	2	Output	Motor Enable	
	3	4			
	5	6			
	7	8	Input	Sector Index	
	9	10	Output	Disk Select 0	
	11	12	Output	Disk Select 1	
	13	14			
	15	16	Output	Motor Enable	Jumpered with E32
GND	17	18	Output	Step Direction	
	19	20	Output	Step Pulse	
	21	22	Output	Write Data	
	23	24	Output	Write Gate	
	25	26	Input	Track 0	
	27	28	Input	Write Protect	
	29	30	Input	Read Data	
	31	32	Output	Side/Head Select	Jumpered by E1, closed for double-sided drive
	33	34	Input	Drive Ready	

4. Firmware Interface

The firmware interface is unsurprisingly very similar to 586T. A notable difference is addition of the Floppy Channel, described below.

The 586T/986T System Reference Manual is not too clear about the offset of various register blocks though they can be for the most part inferred from their respective lengths, since they immediately follow each other.

The layout of register blocks on the 586 is somewhat different due to the inclusion of the *Floppy Drive Interface Registers* block before the *Time of Day Registers*. offsets of each block. It is incidentally also the offset of each command register the firmware looks at, since each block starts with a command register.

Offset	Description
0Ah	Communication Channel Registers (Port 1)
20h	Communication Channel Registers (Port 2)
36h	Communication Channel Registers (Port 3)
4Ch	Communication Channel Registers (Port 4)
62h	Communication Channel Registers (Port 5)
78h	Communication Channel Registers (Port 6)
8Ah	Floppy Drive Interface Registers
D4h	Time of Day Registers

The Communication Channel Registers and Time of Day Registers are described in the 586T/986T System Reference Manual, respectively. The rest of this section deals with the Floppy interface exclusively.

4.1. Floppy Drive Interface Registers

This structure is present at offset 8Ah from the command control block (CCB). Not all members are used by all commands.

Offset	Size	Description	Typical Value
+00h	BYTE	Command	87h, 88h,
+01h	BYTE	Status	00h, 48h, 40h, C0h,
+02h - +04h	BYTE[3]	Next Floppy Command Pointer	Pointer to Floppy Command Queue
			entry
+05h	BYTE	Floppy Command Queue Size	02h
+06h	BYTE	Last Floppy Command Index	01h
+07h	BYTE	Next Floppy Command Index	00h
+08h	BYTE	Unused	00h
+09h	BYTE	Unused	00h
+0Ah - +B3h	BYTE[32]	Floppy 1 Parameters	Floppy Parameters
+B4h - +D3h	BYTE[32]	Floppy 2 Parameters	Floppy Parameters

4.2. Floppy Drive Interface Commands

There factory software has been seen issuing three different command, with two of them having known function:

Command	Description
87h	Set Floppy Drive Parameters
88h	Submit Floppy Command Queue
8fh	Unknown

As with the other command blocks, the most significant bit indicates to the peripheral controller that there's a new command to process. Once the processing is finished, the peripheral controller sets the bit to zero.

4.2.1. Set Floppy Drive Parameters

The Set Floppy Drive Parameters command reads the Floppy Parameters structure (see below) and saves it in the controller's SRAM. The other registers in the block are ignored. This command needs to be called to set the sector size before any Floppy I/O.

Failure to set the sector size up will cause the read commands to overflow a data buffer, corrupting the controller SRAM. The main processor firmware calls this command on startup.

4.2.2. Submit Floppy Command Queue

The Submit Floppy Command Queue command submits the Floppy Command Queue for processing. Said queue is in fact a ring buffer whose entries are the addresses of Floppy Disk Interface Parameter Block structures (described below).

Upon submission, the processing starts with the entry on address in *Next Floppy Command Pointer*. an entry, it increments the next entry index, wrapping it around the queue size if necessary. The processing finishes once the next entry index is equal to the last entry index.

Note that the I/O processor modifies the *Next Floppy Command Index*register when it finishes handlin an entry, but keeps the value in *Next Floppy Command Pointer* unchanged, while still using it to calculate the address of the next entry to process. When you submit another batch of commands When you submit another batch of commands you need to adjust it accordingly.

This command doesn't use the *Floppy Parameters* in the register block, though it requires that the parameters are set by the *Set Floppy Drive Parameters* previously.

4.3. Floppy Parameters

Used to issue I/O commands, not used by command.

Offset	Size	Description	Typical Value
+00h	WORD	Unknown	0200h
+02h	WORD	Sector size	0200h
+04h	BYTE	Unknown	06h
+05h - +1fh	BYTE[]	Unknown/Filler (00h)	00h

4.4. Floppy Command Queue

Offset	Size	Description	Typical Value	
+00h	DWORD	Floppy Command Block Address	Pointer to Floppy Command Block	
+04h	DWORD	Floppy Command Block Address	Pointer to Floppy Command Block	
+04h * n	DWORD	Floppy Command Block Address	Pointer to Floppy Command Block	

4.5. Floppy Disk Interface Parameter Block

Offset	Size	Description	Typical Value	
+00h	High 4 bits	Command	10h, 20h, 1Ah, 2Ah,	
+0011	Low 4 bits	Retries	1011, 2011, 1A11, 2A11,	
+01h	BYTE	Status		
+02h	BYTE	Drive Number	00h	
+03h	BYTE	Track	00h - 28h	
+04h	BYTE	Head	00h, 01h	
+05h	BYTE	Sector	01h – 09h	
+06h - +09h	BYTE[3]	Data Buffer Address		

5. Reading a Floppy from the Firmware ROM Monitor

This section provides a concrete example of using the Floppy Interface of the peripheral controller.

```
FAILED POWER-UP TEST B Controller board error. I got no board.

Monitor Version a2.2

Enter [1] to boot from Hard Disk

Enter [2] to boot from Floppy Disk

Enter [3] to enter Monitor

Enter option: 3
```

First let's find the Channel Control Block address (CCB). Read it from 1FFFCh:

It's 416h. The channel register offsets are calculated from this address. The new command register is at offset 05h from the beginning of the CCB, therefore at address 41Bh.

Now let's prepare a small program that just bumps the command counter so that the controller firmware will know it needs to check for new commands. It then invokes to return to the monitor. We'll place it at address 2000h, which is an arbitrarily chosen address in the DRAM.

It's not strictly necessary here, since each key press results in a serial command and the command registers are checked anyway, but would be necessary for floppy communication outside the monitor.

We can proceed setting the floppy parameters. This is also not strictly necessary if we're using the ROM monitor, because the firmware must have initialized the parameters for us.

Let's do it anyway though, for demonstration purposes. It might also be useful if the floppy format uses a sector of size other than 512 bytes.

The floppy command register is at offset 8Ah from the beginning of the CCB, therefore at address 4A0h.

For reasons explained below, we start setting the bytes from 4A1h and set the command byte (4A0h) last.

```
0000:04A9 00-00
  0000:04AA 00-00
                              Floppy 1 parameters
  0000:04AB 02-02
  0000:04AC 00-00
  0000:04AD 02-02
  0000:04AE 06-06
  0000:04AF 00-00
  0000:04C9 00-00
  0000:04CA 00-00
                              Floppy 2 parameters
  0000:04CB 02-02
  0000:04CC 00-00
  0000:04CD 02-02
  0000:04CE 06-06
  0000:04CF 00-00
  0000:04E9 00-00
  0000:04EA 11-,
< A, B, D, G, I, K, L, M, O, R, S, X > A 0:04A0
  0000:04A0 00-87
                              The "Floppy Parameters" command byte
  0000:04A1 00-,
```

We set the command byte last to make sure the controller firmware doesn't see a valid command until the parameters are all set.

Now we can bump the command counter with the program we entered in at the beginning (though, as stated above, this is not really necessary):

Let's enter some actual floppy I/O requests now.

Start with a seek:

Seek to a different track, so that we're sure we see the head move:

0000:2203	00- 25	Track
0000:2204	00-01	Head
0000:2205	00-05	Sector
0000:2206	00-,	

And then read a sector. The controller will seek back to the appropriate track:

```
< A, B, D, G, I, K, L, M, O, R, S, X > A 0:2300
 0000:2300 00-10
                               Read command
 0000:2301 00-00
                               Status
 0000:2302 00-00
                               Unknown, 00h
 0000:2303 00-05
                               Track
 0000:2304 00-01
                               Head
 0000:2305 00-05
                               Sector
 0000:2306 00-00
                               Destination buffer address: 00002500h
 0000:2307 00-25
 0000:2308 00-00
 0000:2309 00-00
 0000:230A 00-,
```

Create a queue with the three commands we created above:

```
< A, B, D, G, I, K, L, M, O, R, S, X > A 0:2400
  0000:2400 00-00
                                First command at 00002100h
  0000:2401 00-21
                                (See above: seek to track 5)
  0000:2402 00-00
  0000:2403 00-00
  0000:2404 00-00
                                First command at 00002200h
  0000:2405 00-22
                                (See above: seek to track 25h)
  0000:2406 00-00
  0000:2407 00-00
  0000:2408 00-00
                                First command at 00002300h
  0000:2409 00-23
                                (See above: read from CHS 5/1/5 to 2500h)
  0000:240A 00-00
  0000:240B 00-00
  0000:240C 00-,
```

Now invoke the queue the same way as we set the parameters:

```
< A, B, D, G, I, K, L, M, O, R, S, X > A 0:04A1
  0000:04A1 00-ff
                                 Status
  0000:04A2 60-00
                                 The queue is at 00002400h
  0000:04A3 09-24
  0000:04A4 00-00
                                 Queue Length
  0000:04A5 02-04
  0000:04A6 00-03
                                 Last valid command is 03h commands in the queue
  0000:04A7 00-00
                                Start with the command 00h
  0000:04A8 00-00
  0000:04A9 00-00
  0000:04AA 00-,
< A, B, D, G, I, K, L, M, O, R, S, X > \bf A 0:04\bf A0
  0000:04A0 00-c8
                               Now set the command byte
  0000:04A1 48-,
                               to process the queue
```

< A, B, D, G, I, K, L, M, O, R, S, X >

```
< A, B, D, G, I, K, L, M, O, R, S, X > G 0:2000
Break ....
CS:IP FC00:0008 Flags - - - S - A - -
 AX BX CX DX SI DI DS
                                         ES
                                              SS
 The command is in. Let's keep checking the status byte until it indicates success:
< A, B, D, G, I, K, L, M, O, R, S, X > D 0:04A1 1
 0000:04A1
             48
                                               *.H.$.....*
The status is now 48h. We need it to be 40h. Try some more:
< A, B, D, G, I, K, L, M, O, R, S, X > D 0:04A1 1
                                                *.@.$....*
 0000:04A1
Now the status is 40h, which indicates no error.
The sector data is now at 2500h, let's check it out:
< A, B, D, G, I, K, L, M, O, R, S, X > D 0:2500 200
 0000:2500 EB080000 30000000 0002CCEB FD000000 *k...0....lk}...*
 0000:26F0 00000000 00000000 00000000 *.....*
```

5.1. Z80 ROM Monitor

The factory firmware on the IP seems to include some sort of debug monitor capability.

Having a firmware monitor is very helpful for investigation of the system. It would allow controlling the Z80 peripherals directly and operating them without the main 8086 processor.

Unfortunately, I didn't discover a way to enter the factory ROM monitor. I've ended up temporarily replacing the ROM with a different one, allowing me to poke around the controller (whilst the main processor is just stuck in an endless loop).

5.1.1. Custom ROM Monitor

My monitor program is a slight modification² of a Z80 monitor made by MCook for the ZMC80 computer and adjusted for MPF-1 by F.J.Kraan. It is free software (MIT licensed) and fits the purpose perfectly.

The following capture of the ROM monitor output provides some insight into what functionality the monitor provides:

```
Altos 586 IOP Debug Monitor 0.7
<https://github.com/lkundrak/altos586/>
2015 MCook, 2022 F.J.Kraan, 2022 Lubomir Rintel
         Input ? for command list
>?
ZMC80 Monitor Command List
? - view command list
C - clear screen
D - print 100h bytes from specified location
E - edit bytes in memory
F - fill memory range with value
G - jump to memory value
K - call to memory value
M - copy bytes in memory
P - print port scan (00-FF)
R - monitor reset
S - calculate checksum for memory range
Z - dump user registers (STEP)
I - read I/O port
O - write I/O port
: - download intel hex
+ - print next block of memory
- - print previous block of memory
```

² Source code and a binary image are available at https://github.com/lkundrak/Z80SerialMonitor-586/

6. Hard Drive Controller Board

This hard drive controller is based on the Intel 8089 I/O processor. The 8089 processor is connected to on-board logic and sector buffer SRAM via a private bus, allowing them to be accessed without congesting the system bus. It is also connected to the main memory via the system bus, allowing for efficient DMA transfers of the sector data.

The 8089 is controlled with specialized programs placed in the main memory and synchronized with the main CPU with interrupts.

The firmware includes one such program to handle the disk reads and writes. It is parametrized by I/O parameter blocks (IOPB) placed in the main memory. It is described in detail in section below.

So far, I've observed two I/O programs for the controller board: one in the firmware and another on a diagnostic floppy. The Xenix operating system apparently uses yet another -- I haven't looked into that one yet.

The following paragraphs give a quick overview of the two I/O programs followed by commented disassembly listings. The programs can be assembled using Eric Smith's Assembler for the 8089.³

6.1. Firmware IO Handling Program

This is the simple of the two I/O programs. It is provided in the firmware image and run on the 8089 on the hard drive board to service the I/O operations.

It is controlled by the I/O parameter block structure to ultimately transfer data between a main memory buffer and the hard drive. The data is not transferred directly between the controller's data port and the main memory. Instead, the data is bounced through a a buffer in controller board's SRAM.

This utilizes the main memory bus more efficiently. The controller data port transfers eight bits of data at a time at a rate determined by the disk rotation speed. These slow accesses are done over a bus that's private to the controller board, avoiding congestion of the main bus. On the other hand, the SRAMs allows 16-bit accesses at a much faster rate, utilizing the main memory as little as possible.

On reads, the data is first transferred from the data port to the SRAM buffer and then from SRAM to the main memory. On writes, the data is copied from main memory to SRAM buffer and then from SRAM to the data port.

The number of sectors specified in IOPB determines how many transfers are done between the data port and the SRAM. The number of bytes transferred between SRAM and main memory is specified in IOPB directly. These two numbers are related -- main memory transfer needs to transfer sufficient number of bytes to include complete sector data.

The I/O program listing⁴ follows.

³ Available from https://github.com/brouhaha/i89

⁴ Assembly source available from https://github.com/lkundrak/altos586/blob/main/8089/iop3268.asm

```
; Program parameters, passed by the user.
IOPB
                 STRUC
                 DS 4
                          ; Task Block pointer
                DS 1
IOPB_OP:
IOPB_STATUS:
                DS 1
IOPB_CYL:
                DS 2
IOPB_DRV_HD:
                DS 1
                DS 1; Starting sector
IOPB_SEC:
IOPB_BYTE_CNT: DS 2
                         ; The DMA buffer size
IOPB_DMA_BUF: DS 4
                         ; The DMA b uffer address
IOPB_SEC_CNT: DS 2 ; Number of sector s
                         ; The following members are reserved for use by the IOP.
IOPB IO SIZE: DS 2 ; Written by the IOP
IOPB_LAST_CYL: DS 2
                         ; Last cylinder we've seeked to
IOPB_LINK_REG: DS 2
                          ; Return address of subroutine calls
IOPB
                 ENDS
i Bits of IOPB_OP.
OP BIT RD
                EQU 0
                          ; Read (0) or write (1)
OP_BIT_FORMAT EQU 2
OP_BIT_ECC
                 EQU 3
                          ; Include the ECC bits on read
OP BIT SEL
                 EQU 5
                          ; Drive/Head has changed
OP_CMD_BITS
                 EQU OFH
; Bits of IOPB_STATUS.
STATUS_BIT_ERR EQU 7
; These are addresses in 8089's I/O space.
HDD SRAM ADDR
                 EQU 00000H
HDD_REGS_ADDR EQU OFFDOH
; Registers at HDD_REGS_ADDR.
PORT_DATA
                EQU 00H
PORT DRV HD
                EQU 02H
PORT 04H
                EQU 04H
PORT_CMD_STAT EQU 06H
```

```
; The entry point.
ENTRY:
                  movi qc, HDD REGS ADDR
                  ; Is bit 5 set?
                  jnbt [pp].IOPB_OP,OP_BIT_SEL,NO_SEL
                  ; If bit 5 is set, then drive and head might have
                  ; Changed and we need ensure correct head is selected.
                  movbi [gc].PORT_CMD_STAT,80H
                  movb [gc].PORT_DRV_HD,[pp].IOPB_DRV_HD
в0:
                  jnbt [gc].PORT_CMD_STAT,7,B0
                  movbi [qc].PORT CMD STAT, 20H
                  movi [pp]. IOPB_LAST_CYL, 0
C0:
                         [gc].PORT_CMD_STAT,0,C0
                  jbt
                  ljnbt [gc].PORT_04H,0,RETURN_81H; Error?
NO_SEL:
                  andbi [pp].IOPB_OP,OP_CMD_BITS ; Command bits mask
                  ljzb [pp]. IOPB_OP, RETURN_00H; Command is zero
                  ; Select head
                  movb [gc].PORT_DRV_HD,[pp].IOPB_DRV_HD
B1:
                  jnbt [gc].PORT_CMD_STAT,7,B1
                  ; Seek to cylinder
                  movb [gc],[pp].IOPB_LAST_CYL; [gc].PORT_DATA
                  movb [gc],[pp].IOPB_LAST_CYL+1;[gc].PORT_DATA
                  movb [gc].PORT_04H,[pp].IOPB_CYL
                  \label{eq:cyl} \texttt{movb} \quad \texttt{[gc].PORT\_04H,[pp].IOPB\_CYL} + 1
                  movbi [qc].PORT CMD STAT, 10H
                  mov
                         [pp].IOPB_LAST_CYL,[pp].IOPB_CYL
C1:
                         [gc].PORT_CMD_STAT,0,C1
                  jbt
D0:
                  jnbt [gc].PORT_04H,1,D0
                  ; Zero sectors? Nothing to do.
                         [pp].IOPB_SEC_CNT,RETURN_00H
                  jzb
                  ; A read?
                  jbt
                         [pp].IOPB_OP,OP_BIT_RD,HDD_SRAM_XFER
                  ; Whis is a write. Transfer the data from main memory
                  ; DMA buffer to controller board SRAM first
                         ga,[pp].IOPB_DMA_BUF
                  lpd
                  movi
                        gb, PORT_DATA
                  mov
                         bc,[pp].IOPB_BYTE_CNT
                  call [pp].IOPB_LINK_REG, MEM_SRAM_XFER
```

```
; Transfer between SRAM and the HDD's data register.
; Common to read and write.
HDD_SRAM_XFER:
                  movi gb, HDD_SRAM_ADDR
                  movi mc, OFE80H; Why?
                  movi ga, HDD_REGS_ADDR+PORT_DATA
                  movi [pp].IOPB_IO_SIZE,512
                  ; A read or write?
                  jnbt [pp].IOPB_OP,OP_BIT_RD,XFER_WR
                  movi cc, 8A28H; Ar ead. Setup for SRAM to HDD transfer.
                  wid
                         8,16
                  ; Add 5 bytes if ECC requested.
                         [pp].IOPB_OP,OP_BIT_ECC,XFER_SEC
                         [pp].IOPB_IO_SIZE,512+5
                  movi
                  jmp
                         XFER_SEC
                  ; A write. Setup for transfer from SRAM to HDD.
XFER_WR:
                  movi cc,5628H
                  wid
                         16,8
                  jnbt [pp].IOPB OP,OP BIT FORMAT,XFER SEC
                  movi [pp].IOPB_IO_SIZE, 4; Sector header
                  ; Transfer one sector
XFER SEC:
                        bc,[pp].IOPB_IO_SIZE
                  mov
                        [gc],[pp].IOPB_SEC;[gc].PORT_DATA
                  movb
                  xfer
                  ; Check for errors
                  movb [gc].PORT_CMD_STAT,[pp].IOPB_OP
                  jmcne [qc].PORT CMD STAT, RETURN ERR
                  ; Are we done or there's more?
                  decb [pp].IOPB_SEC_CNT
                         [pp].IOPB_SEC_CNT,XFER_DONE
                  jzb
                  incb [pp]. IOPB_SEC; There's more.
                         XFER_SEC
                  jmp
                  ; HDD - SRAM transfer done. We're done if it was a write.
XFER DONE:
                  jnbt [pp].IOPB_OP,OP_BIT_RD,RETURN_00H
                  ; This was a read. We need to transfer from SRAM to DMA buffer.
                  lpd
                         gb, [pp]. IOPB_DMA_BUF
                  movi ga, PORT_DATA
                  mov
                         bc, [pp]. IOPB_BYTE_CNT
                  call
                        [pp].IOPB_LINK_REG, MEM_SRAM_XFER
                  ; Successful return.
RETURN_00H:
                  movbi [pp].IOPB_STATUS,0
                        DONE
                  jmp
```

```
; A subroutine that does a memory-to memory transfer.
; Used to copy data between the controller SRAM and the DMA buffer in
; main memory, in either direction.
MEM_SRAM_XFER:
                         16,16
                 wid
                  movi cc,0C208H
                  xfer
                  nop
                  movp tp,[pp].IOPB_LINK_REG
; Error return. Communicate the error from the status.
; register to the IOPB.
RETURN_ERR:
                  movb [pp].IOPB_STATUS,[gc].PORT_CMD_STAT
                  andbi [pp]. IOPB_STATUS, 7EH
                  setb [pp].IOPB_STATUS,STATUS_BIT_ERR
                  movbi [gc].PORT_CMD_STAT,00H
                         DONE
                   jmp
                  ; Error 1.
RETURN 81H:
                  movi [pp].IOPB_STATUS,(1<<STATUS_BIT_ERR)+1</pre>
DONE:
                  sintr
                  hlt
```

6.2. Diagnostic IO Handling Program

This one is from the diagnostic utility floppy program, that's capable of reading, writing, verifying and formatting the disk.

It is designed for high-performance I/O. The diagnostic program first invokes a routine that copies another I/O program into the controller board SRAM. That way it can be executed without contending for the main memory.

It is also able to enqueue two I/O requests, presumably for submitting requests to two drives concurrently.

The I/O program listing⁵ follows.

```
; These are addresses in 8089's I/O space.
HDD_SRAM_ADDR
                 EQU 00000H
LOADER_OFFSET EQU 00400H
HDD_REGS_ADDR EQU OFFDOH
; Registers at HDD_REGS_ADDR.
REG DATA
                  EQU 00H
                  EQU 02H
REG_DRV_HD
                  EQU 04H; Write cylinder, read seek status
REG_CYL_SK
REG_CMD_STAT
                  EQU 06H; Write command, read op status
REG_28H
                  EQU 28H
; Parameter Block
;-
PB
                   STRUC
PB TB:
                   DS 4
                            ; Offset:Segment of first command block
PB_IN1_ADDR:
                  DS 4
PB_IN2_ADDR:
                            ; Offset:Segment of second command block
                  DS 4
PB_OUT1_ADDR:
                           ; Of fset:Segment of first result block
                   DS 4
PB OUT2 ADDR:
                   DS 4
                           ; Of fset:Segment of second result block
PB
                   ENDS
```

 $^{^5 \} Assembly \ source \ available \ from \ \textit{https://github.com/lkundrak/altos586/blob/main/8089/hd-test.asm}$

```
; Command Block.
                STRUC
CB_OP:
                DS 1
CB_STAT:
               DS 1
CB_CYL_LO:
               DS 1
               EQU CB_CYL_LO
CB_CYL
CB_CYL_HI:
               DS 1
CB_DRV_HD:
               DS 1
CB_SEC:
                DS 1
CB_LEN:
                DS 2
               DS 4
CB_BUF_ADDR:
CB_REG_28H:
               DS 1
CB NUM SECS:
                DS 1
CB_RETRIES:
                DS 1
                DS 1
CB_SIZE:
CB
                ENDS
; Value following the Command Block in the Working Area.
WA
                STRUC
                DS CB_SIZE
WA_CUR_CYL:
                DS 2
WA_CB_ADDR:
              DS 4; Original addr
                                   ess of this CB
WA_LINK:
               DS 4
WA_LINK1:
               DS 4
WA_XFER_LEN:
               DS 2
WA_TMP_CMD:
               DS 1
WA_CUR_DRV_HD: DS 1
WA LINK2:
                DS 4
WA_ERR_TRIES:
                DS 1
WA_ERR_SEC:
                DS 1
WA_SIZE:
WA
                ENDS
i Bits of CB_OP.
OP_BIT_RD
                EQU 0
                       ; Read (0) or write (1)
OP_BIT_FMT
                EQU 2 ; Format
OP BIT ECC
                EQU 3
                       ; Include the ECC bits on read
                        ; Drive/Head has changed
OP_BIT_SEL
                EQU 5
OP_BIT_UNK6
                EQU 6
OP_BIT_UNK7
                EQU 7
```

```
; The entry point.
CODE:
                lpd
                        ga,[pp].PB_IN1_ADDR
                jzb
                        [ga].CB_OP,CH1_CHECKED
                movi
                        gb,WA1
               movi
                        qc,WA1
                lcall
                       [gc].WA_LINK,COPY_CB
CH1_CHECKED:
                lpd
                        ga,[pp].PB_IN2_ADDR
                jzb
                        [ga].CB_OP,CH2_CHECKED
                movi
                       gb,WA2
               movi
                       qc, WA2
                lcall
                       [gc].WA_LINK,COPY_CB
CH2_CHECKED:
                        ga, HDD_REGS_ADDR
               movi
               movi
                       gc,WA1
                jzb
                       [gc].CB_OP,CH1_SEEKED
                lcall
                       [gc].WA_LINK,SET_TRACK
                        [gc].CB_OP,OP_BIT_SEL,SEEK_CH1
                jnbt
                        [gc].WA_LINK1,SEL_DRIVE
                lcall
                jmp
                        CH1 SEEKED
SEEK CH1:
                lcall [gc].WA_LINK1,SEEK_TO_CYL
CH1_SEEKED:
               movi
                        gc, WA2
                       [gc].CB_OP,CH2_SEEKED
                jzb
                lcall
                        [gc].WA_LINK,SET_TRACK
                jnbt
                       [qc].CB OP,OP BIT SEL,SEEK CH2
                lcall
                       [gc].WA_LINK1,SEL_DRIVE
                       CH2_SEEKED
                jmp
SEEK CH2:
                lcall [gc].WA_LINK1,SEEK_TO_CYL
CH2 SEEKED:
               movi
                        gc,WA1
                jzb
                        [gc].CB_OP,CH1_DONE
                jbt
                        [gc].CB_OP,OP_BIT_SEL,CH1_OP_DONE
                lcall [gc].WA_LINK,SET_TRACK
                       [gc].CB_OP,OP_BIT_RD,CH1_WAIT_SEEK
                jbt
                lcall
                       [qc].WA LINK, COPY FROM DMA
CH1_WAIT_SEEK:
                jnbt
                        [ga].REG_CYL_SK,1,CH1_WAIT_SEEK
                        [gc].WA_LINK2,DO_DISK_CMD
                lcall
                jnbt
                        [gc].CB_OP,OP_BIT_RD,CH1_OP_DONE
                lcall
                        [gc].WA_LINK,COPY_TO_DMA
CH1_OP_DONE:
               movi
                        ga,WA1
                lpd
                        gb,[pp].PB_OUT1_ADDR
                lcall [gc].WA_LINK,COPY_CB
CH1_DONE:
```

```
movi
                         gc, WA2
                jzb
                         [gc].CB_OP,CH2_DONE
                         [gc].CB_OP,OP_BIT_SEL,CH2_OP_DONE
                 jbt
                lcall
                         [qc].WA LINK, SET TRACK
                jbt
                         [gc].CB_OP,OP_BIT_RD,CH2_WAIT_SEEK
                lcall
                         [gc].WA_LINK,COPY_FROM_DMA
                         [ga].REG_CYL_SK,1,CH2_WAIT_SEEK
CH2_WAIT_SEEK:
                jnbt
                        [gc].WA_LINK2,DO_DISK_CMD
                lcall
                         [gc].CB_OP,OP_BIT_RD,CH2_OP_DONE
                jnbt
                lcall
                         [gc].WA_LINK,COPY_TO_DMA
CH2_OP_DONE:
                movi
                         ga,WA2
                lpd
                         gb, [pp].PB_OUT2_ADDR
                lcall [gc].WA_LINK,COPY_CB
CH2 DONE:
                sintr
                hlt
; Transfer data between the controller SRAM buffer and the controller.
; Handles the situations when the transfer crosses track boundary too.
                         [qc].CB NUM SECS, DISK CMD DONE
DO DISK CMD:
                ljzb
                movb
                         ix,[gc].CB_OP
                andi
                         ix,7h
                ljz
                         ix, DISK CMD DONE
                movbi
                        [gc].WA_ERR_SEC,0ffh
                movi
                        gb, HDD_SECTOR_BUF
                movi
                        [gc].WA_XFER_LEN,512
                jnbt
                        [gc].CB_OP,OP_BIT_RD,SETUP_WRITE
                movbi [gc].WA_TMP_CMD,1h
                         [gc].CB_OP,OP_BIT_ECC,SETUP_READ
                jnbt
                movbi
                         [qc].WA TMP CMD,9h
                addi
                        [gc].WA_XFER_LEN,5h; ECC
SETUP_READ:
                movi
                         cc,8a28h
                wid
                         8,16
                         SETUP_XFER
                jmp
SETUP_WRITE:
                movbi
                         [gc].WA_TMP_CMD,2h
                movi
                         cc,5628h
                wid
                         16,8
                jnbt
                        [gc].CB_OP,OP_BIT_FMT,SETUP_XFER
                movbi
                        [qc].WA TMP CMD,4h
                movi
                        [gc].WA_XFER_LEN,4h
SETUP_XFER:
                movi
                         mc,0fe80h
DO_XFER:
                movb
                         [ga],[gc].CB_SEC
                movp
                         [gc].WA_CB_ADDR,gb
                mov
                         bc,[gc].WA_XFER_LEN
                xfer
                         [ga].REG_CMD_STAT,[gc].WA_TMP_CMD
                movb
                         [ga].REG_CMD_STAT,DISK_OP_ERROR
                jmcne
```

```
decb
                         [gc].CB_NUM_SECS
                 jzb
                         [gc].CB_NUM_SECS,DISK_CMD_DONE
                ; Next sector on same track
                 incb
                        [qc].CB SEC
                 jnbt
                         [gc].CB_SEC,4,DO_XFER
                ; Next head
                incb
                        [gc].CB_DRV_HD
                        mc,[gc].WA_CUR_DRV_HD
                movb
                ori
                        mc,0f00h
                 jmce
                        [gc].CB_DRV_HD,NEXT_CYL
SECTOR_ZERO:
                movb
                        [ga].REG_DRV_HD,[gc].CB_DRV_HD
                movbi [gc].CB_SEC,0h
                         SETUP XFER
                 qmj
                ; Next cylinder
NEXT_CYL:
                inc
                        [gc].CB_CYL
                lcall
                         [gc].WA_LINK1,SEEK_TO_CYL
                andbi [gc].CB_DRV_HD,0f0h
POLL_OP_SEEK:
                jnbt
                        [ga].REG_CYL_SK,1,POLL_OP_SEEK
                         SECTOR_ZERO
                jmp
DISK CMD DONE:
                jbt
                        [ga].REG_CMD_STAT,0,DISK_CMD_DONE
                movp
                         tp,[gc].WA_LINK2
i Retry on errors.
DISK_OP_ERROR:
                 jbt
                         [ga].REG_CMD_STAT,5,ERR_STATUS
                 jbt
                         [ga].REG_CMD_STAT,1,ERR_STATUS
                        [gc].CB_OP,OP_BIT_UNK6,NO_NEED_RECAL
                 jbt
                movb
                         mc,[gc].CB_SEC
                ori
                        mc,0ff00h
                 jmce
                        [gc].WA_ERR_SEC,ERR_SEC_SET
                movb
                        [gc].WA_ERR_SEC,[gc].CB_SEC
                movbi
                        [gc].WA_ERR_TRIES,0h
ERR_SEC_SET:
                         [qa]. REG CMD STAT, 2, SKIP RECAL
                 jbt
                 jbt
                         [ga].REG_CMD_STAT,4,SKIP_RECAL
                         [\, \verb"gc"] \, . \, \verb"WA\_ERR\_TRIES", \verb"GIVE\_UP"
                 jnzb
                ; Seek to cylinder zero to recalibrate
                movbi [ga].REG_CYL_SK,0
                lcall
                        [gc].WA_LINK1,SEL_DRIVE
                lcall [qc].WA LINK1, SEEK TO CYL
POLL RECAL:
                jnbt
                        [ga].REG_CYL_SK,1,POLL_RECAL
                 jmp
                         RECAL_DONE
SKIP RECAL:
                 jbt
                        [gc].WA_ERR_TRIES,4,GIVE_UP
RECAL_DONE:
                incb
                        [gc].CB_RETRIES
                incb
                        [gc].WA_ERR_TRIES
NO_NEED_RECAL:
                movb
                        [gc].CB_STAT,[ga].REG_CMD_STAT
                andbi
                         [gc].CB_STAT,7eh
                movbi
                        [ga].REG_CMD_STAT,0h
```

```
; Try again
                 ljnbt
                         [gc].CB_OP,OP_BIT_UNK6,SETUP_XFER
GIVE UP:
                 setb
                          [gc].CB_STAT,7
                 movp
                          tp,[gc].WA_LINK2
; Communicate the error and cancel both channels.
ERR_STATUS:
                          [gc].CB_STAT,[ga].REG_CMD_STAT
                 movb
                 ljmp
                          ERR_FINISH
; Copy the Command Block between the controller SRAM and main memory.
COPY CB:
                 movp
                          [gc].WA_CB_ADDR,ga
                 movi
                          bc, CB_SIZE
                 call
                         [gc].WA_LINK1,MEMCPY_16
                       ga,[gc].WA_CB_ADDR
                 movp
                 movbi [ga].CB_OP,0h
                 movi ga,HDD_REGS_ADDR
                         [ga].REG_CMD_STAT,1,ERR_STATUS
                 jbt
                 movp
                          tp,[gc].WA_LINK
; Copy a block of data (sector, CB) between main memory and SRAM.
MEMCPY_16:
                 wid
                          16,16
                 movi
                          cc,0c008h
                 xfer
                 nop
                 movp tp,[gc].WA_LINK1
; Set head and a cylinder.
SET_TRACK:
                 movb
                          [ga].REG_DRV_HD,[gc].CB_DRV_HD
                 movb
                          [ga].REG_CYL_SK,[gc].WA_CUR_CYL+0
                 movb
                          [ga]. \textbf{REG\_CYL\_SK}, [gc]. \textbf{WA\_CUR\_CYL} + 1
                          [gc].CB_STAT,0h
                 movbi
                 ; Wait for bit 7 to come up for up to FFh attempts
                 movi bc,0h
                 dec
POLL_TRK_SET:
                          bc
                 jz
                          bc, ERR FINISH
                 jnbt
                         [ga].REG_CMD_STAT,7,POLL_TRK_SET
                 movp tp,[gc].WA_LINK
```

```
; Seek to a cylinder.
SEEK TO CYL:
                movb
                         [qa],[qc].WA CUR CYL+0
                movb
                        [ga],[gc].WA_CUR_CYL+1
                movb
                        [ga].REG_CYL_SK,[gc].CB_CYL_LO
                movb
                        [ga].REG_CYL_SK,[gc].CB_CYL_HI
                movbi [ga].REG_CMD_STAT,10h
                        [gc].WA_CUR_CYL,[gc].CB_CYL
                mov
POLL_SEEK:
                        [ga].REG_CMD_STAT,0,POLL_SEEK
                jbt
                movp
                        tp,[gc].WA_LINK1
; Select a disk drive unit.
SEL_DRIVE:
                movbi
                        [ga].REG_CMD_STAT,80h
                movbi
                        [gc].CB_STAT,0h
                jnbt
                        [gc].CB_OP,OP_BIT_UNK7,NO_BIT_7
                        [gc].WA_CUR_DRV_HD,[gc].CB_DRV_HD
                movb
NO_BIT_7:
                movb
                        [ga].REG_DRV_HD,[gc].CB_DRV_HD
                ; Wait for bit 7 to come up for up to FFh attempts
                movi
                        bc,0h
POLL_STAT_7:
                dec
                        bc
                        bc, ERR_FINISH
                jz
                jnbt
                        [ga].REG_CMD_STAT,7,POLL_STAT_7
                movbi [ga].REG_CMD_STAT, 20h
                movi
                        [gc].WA_CUR_CYL,0h
                        [ga].REG_CMD_STAT,0,POLL_SEL
POLL SEL:
                jbt
                        tp,[gc].WA_LINK1
                movp
; Indicate errors in both channels.
ERR_FINISH:
                movi
                        ga,WA1
                orbi
                        [ga].CB_STAT,81h
                lpd
                        gb,[pp].PB_OUT1_ADDR
                lcall
                        [gc].WA_LINK,COPY_CB
                movi
                        ga,WA2
                orbi
                        [ga].CB_STAT,81h
                lpd
                       gb,[pp].PB_OUT2_ADDR
                lcall [gc].WA_LINK,COPY_CB
                sintr
                hlt
```

```
; Copy data indicated by CB from main memory to controller SRAM.
COPY FROM DMA:
                 jz
                          [qc].CB LEN,COPIED IN
                          [gc].CB_REG_28H,4
                 setb
                          [ga].REG_28H,[gc].CB_REG_28H
                 movb
                 ; Copy from DMA buffer to sector buf
                          ga,[gc].CB_BUF_ADDR
                 lpd
                 movi
                          {\tt gb} , {\tt HDD\_SECTOR\_BUF}
                 mov
                          bc,[gc].CB_LEN
                 lcall [gc].WA_LINK1,MEMCPY_16
                          ga, HDD_REGS_ADDR
                 movi
                 movbi [qa].REG 28H,10h
                 ljbt
                         [ga].REG_CMD_STAT,1,ERR_STATUS
COPIED_IN:
                 movp
                          tp,[gc].WA_LINK
; Copy data indicated by CB to main memory from controller SRAM.
COPY_TO_DMA:
                 jz
                          [gc].CB_LEN,COPIED_OUT
                          [qc].CB REG 28H,4
                 setb
                 movb
                          [ga].REG_28H,[gc].CB_REG_28H
                 ; Copy sector buf to DMA buffer
                 lpd
                          gb,[gc].CB_BUF_ADDR
                 movi
                          ga, HDD_SECTOR_BUF
                 mov
                        bc,[gc].CB_LEN
                 lcall [gc].WA_LINK1,MEMCPY_16
                 movi
                          ga, HDD_REGS_ADDR
                 movbi
                         [qa].REG 28H,10h
COPIED OUT:
                 movp
                          tp,[gc].WA_LINK
; Command block copies and sector data copies in controller SRAM.
WA1:
                 DSWA_SIZE
WA2:
                 DSWA_SIZE
HDD_SECTOR_BUF:
```

```
; Pad the loader to 0400h
DS LOADER_OFFSET-HDD_SECTOR_BUF

; The bootstrap routine. Copies the IOP to controller SRAM so that it
; can be executed without main memory contention.
;
```

LOADER:

```
lpd
       gc,[pp];[pp].PB_TB
lpd
        ga,[pp];[pp].PB_TB
movi
        gb, HDD_SRAM_ADDR
; Move 400 bytes back to set BC=IOP base
movi bc, LOADER
not
        bc
inc
       bc
; Initialize CH2 area
       [gb],bc
mov
add
       ga,[gb]
add
      gc,[gb]
addi gc, WA2
      gb,[pp].PB_IN2_ADDR
lpd
movb
       [gc].WA_CUR_DRV_HD,[gb].CB_DRV_HD
movi
      gb, HDD_SRAM_ADDR
; Initialize CH1 area
movp [gb],ga
movp gc,[gb]
addi gc, WA1
lpd
      gb,[pp].PB_IN1_ADDR
movb [gc].WA_CUR_DRV_HD,[gb].CB_DRV_HD
movi
       qb, HDD SRAM ADDR
; Relocate the IOP
movi
      bc, HDD_SECTOR_BUF
lcall
       [gc].WA_LINK1,MEMCPY_16
movi
        ga, HDD_REGS_ADDR
movbi
       [ga].REG_28H,10h
sintr
hlt
```